
Royal St. George’s College		Design Engineering Studio
	KeypadScanning.ino
Advanced Computer Engineering School	ICS3U-E: AVR Foundations

// PROJECT  :KeypadScanning
// PURPOSE  :Present the scanning (aka polling) algorithm behind the Keypad library
// COURSE   :ICS3U
// AUTHOR   :Modified from Arduino Forum post
// DATE     :2023 03 01
// MCU      :328P
// STATUS   :Working  (need to debounce...)
// REFERENCE:https://forum.arduino.cc/t/keypad-without-keypad-library/656198/7
// REFERNCE :https://www.adafruit.com/product/1824

#define C1  10        //Arduino digital pins
#define C2  9
#define C3  8
#define R1  7
#define R2  6
#define R3  5
#define R4  4

#define PAUSE 100

uint8_t pinCols[3] = {C1, C2, C3};        //pins of the different columns
uint8_t pinRows[4] = {R1, R2, R3, R4};    //pins of the different rows
uint8_t numRows = sizeof(pinRows);
uint8_t numCols = sizeof(pinCols);

int8_t trueCol;                          //active key pressed
int8_t trueRow;                          //

char keys[4][3] = {                //correspondence with the keys of the keypad
  {'1', '2', '3'},
  {'4', '5', '6'},
  {'7', '8', '9'},
  {'*', '0', '#'}
};


void setup() {
  Serial.begin(9600);
  while (!Serial);

  for (uint8_t c = 0; c < numCols; c++) {       //Columns as HIGH output pins
    pinMode(pinCols[c], OUTPUT);
    digitalWrite(pinCols[c], HIGH);
  }
  for (uint8_t r = 0; r < numRows; r++) {       //Rows as input pins
    pinMode(pinRows[r], INPUT);
  }
}

void loop() {
  readKey();                                //scan the matrix
  if (trueRow >= 0 && trueCol >= 0)         //anything?
    Serial.println(keys[trueRow][trueCol]); //if so, display the active key
  delay(PAUSE);
}

void readKey() {
  trueRow = -1;                                 //assume nothing
  trueCol = -1;                                 //
  for (uint8_t i = 0; i < numRows; i++) {
    if (digitalRead(pinRows[i])/* == HIGH*/) {  //read each row for a HIGH
      trueRow = i;                              //identified the row! Now,
      for (uint8_t x = 0; x < numCols; x++) {   //present LOW on all columns
        for (uint8_t c = 0; c < numCols; c++)   //but 1 while reading the true row
          digitalWrite(pinCols[c], LOW);        //to see what comes through...
        digitalWrite(pinCols[x], HIGH);         //
        //if true row is HIGH we have also discovered the column!
        if (digitalRead(pinRows[trueRow])/* == HIGH*/) {
          trueCol = x;
        }
      }
    }
  }
  //restore columns to HIGH
  for (uint8_t c = 0; c < numCols; c++) {
    pinMode(pinCols[c], OUTPUT);
    digitalWrite(pinCols[c], HIGH);
  }

